wfopen  "C:\Users\Maryam\Desktop\BS Studies\PhD Thesis-II\EViews and STATA Progarm Codes\Chapter-8\Singapore.wf1"
'******************************************************************
'Group Plots for Real Exchange Rate (NT) & Productivity Gap
'******************************************************************
group gA rer_def_NT A_tilde
freeze(group_plot) gA.line(x)
group_plot.setelem(1) lcolor(black) symbol(7) lpat(1)
group_plot.setelem(2) lcolor(black) symbol(4) lpat(1)
group_plot.setelem(3) lcolor(black) symbol(1) lpat(1)
group_plot.setelem(3) lcolor(black)
group_plot.options linepat
group_plot.addtext(t) Real Exchange Rate (NT-based) and Productivity Gaps (Singapore & U.S): 1970-2006
group_plot.addtext(b) Year
group_plot.addtext(l) PNT  
group_plot.addtext(l) NER 
group_plot.addtext(l) rer_def_NT
group_plot.addtext(r) A_TILDE
'************************************************************
'************************************************************

create y 1970 2006
'importing data from Excel for Singapore
import  "C:\Users\Maryam\Desktop\BS Studies\PhD Thesis-II\EViews and STATA Progarm Codes\Chapter-8\Chapter 8.xlsx" range="Singapore"

'**************************************************************************************************
'VERIFYING THE ASSUMPTION OF PPP FOR Singapore AND U.S. TRADABLES PRICES
'**************************************************************************************************
'*************************************************************
'STEP 0: Tests for Unit Root in Individual Time Series
'*************************************************************

'***************************************************************
'Graph for Singapore's  pT
'***************************************************************
                                        
genr pT = pT
freeze(figure_pT) pT.line
figure_pT.addtext(t) pT (Singapore):  1970-2006
figure_pT.addtext(b) Year
figure_pT.addtext(l) pT
figure_pT.legend(off)
                                                 
'We see from the FIGURE that pT has a time trend to it.  So we would include both an intercept and a time trend in our unit root regression equtions. 

'*********************************************************************************
'ADF Unit Root Test for Singapore's pT
'We now run our first ADF test
'**********************************************************************************
 
freeze(table_8_2_pT_adf) pT.uroot(adf,trend,info=sic)

'Note that the SIC automatic lag selection picks lags, p = 0.  The unit root test produces a t-value of -2.14 which is greater than our 5% criterion -3.54.  Thus, at this point, we cannot reject the null of a unit root.

'Now, let's check for white noise.  To do that, I first set all the residuals = 0, then run the adf test and finally will check for white noise.

genr resid = 0
freeze(mode=overwrite,pT_adf) pT.uroot(adf,const,trend,info=sic)
freeze(pT_adf_correl) resid.correl
 
'Based on the Q-statistic, I conclude that the residuals are white noise.  Putting it all together, I conclude that the pT series is not level stationary.

'The next thing I do is test whether the differenced series is stationary using the ADF test.  I once again begin by graphing the (differenced) series.
 
genr pTdiff = d(pT)
freeze(figure_pTdiff) pTdiff.line
figure_pTdiff.addtext(t) DpT (Singapore):  1970-2006
figure_pTdiff.addtext(b) Year
figure_pTdiff.addtext(l) DpT
figure_pTdiff.legend(off)

'The graph is not particularly illuminating.  Depending on how you look at it, it could have a time trend to it.  However, since the previous ADF test concluded that the equation for pT do have a time trend, we run the ADF test with both a constant as well as a time trend.

'So we begin the whole process over again: 

genr pTdiff = d(pT)
freeze(table_8_2_pTdiff1_adf) pTdiff.uroot(adf,const,info=sic)

'Note that the SIC automatic lag selection picks no lags, p =0.  The unit root test produces a t-value of -3.70 which is smaller our 5% criterion -2.95.  Thus, we may now reject the null of nonstationarity in first differenced series of pT.  There is no reason to go further.  The last thing we do is check for white noise.

genr resid = 0
freeze(mode=overwrite,pTdiff1_adf) pTdiff.uroot(adf,const,trend,info=sic)
freeze(pTdiff1_adf_correl) resid.correl

''Based on the Q-statistic, I conclude that the residuals are white noise.  Putting it all together, I conclude that the pT series is I(1).

'*********************************************************************************
'DF-GLS Unit Root Test for Singapore's pT
'We now run our first dfgls test
'**********************************************************************************
 
freeze(table_8_2_pT_dfgls) pT.uroot(dfgls,trend,info=sic)

'Note that the SIC automatic lag selection picks lags, p = 0.  The unit root test produces a t-value of -1.13 which is greater than our 5% criterion -3.19.  Thus, at this point, we may not reject the null of a unit root.

'Now let's see if the series is difference stationary or not

genr pTdiff = d(pT)
freeze(table_8_2_pTdiff1_dfgls) pTdiff.uroot(dfgls,const,info=sic)

'Note that the SIC automatic lag selection picks no lags, p = 0.  The unit root test produces a t-value of -3.76 which is smaller than our 5% criterion -1.95.  Thus, we may now reject the null of nonstationarity in first differenced series of pT.  
 
''Putting it all together, I conclude that the pT series is I(1), a finding compatable with my ADF test results.

'***************************************************************
'Graph for Singapore's  pT_us_PPP
'***************************************************************
                                        
genr pT_us_PPP = pT_us_PPP
freeze(figure_pT_us_PPP) pT_us_PPP.line
figure_pT_us_PPP.addtext(t) pT_us_PPP (Singapore):  1970-2006
figure_pT_us_PPP.addtext(b) Year
figure_pT_us_PPP.addtext(l) pT_us_PPP
figure_pT_us_PPP.legend(off)
                                                 
'We see from the FIGURE that pT_us_PPP has a time trend to it.  So we would include both an intercepT_us_PPP and a time trend in our unit root regression equtions. 

'*********************************************************************************
'ADF Unit Root Test for Singapore's pT_us_PPP
'We now run our first ADF test
'**********************************************************************************
 
freeze(table_8_2_pT_us_PPP_adf) pT_us_PPP.uroot(adf,trend,info=sic)

'Note that the SIC automatic lag selection picks lags, p = 1.  The unit root test produces a t-value of -1.53 which is greater than our 5% criterion -3.54.  Thus, at this point, we cannot reject the null of a unit root.

'Now, let's check for white noise.  To do that, I first set all the residuals = 0, then run the adf test and finally will check for white noise.

genr resid = 0
freeze(mode=overwrite,pT_us_PPP_adf) pT_us_PPP.uroot(adf,const,trend,info=sic)
freeze(pT_us_PPP_adf_correl) resid.correl
 
'Based on the Q-statistic, I conclude that the residuals are white noise.  Putting it all together, I conclude that the pT_us_PPP series is not level stationary.

'The next thing I do is test whether the differenced series is stationary using the ADF test.  I once again begin by graphing the (differenced) series.
 
genr pT_us_PPPdiff = d(pT_us_PPP)
freeze(figure_pT_us_PPPdiff) pT_us_PPPdiff.line
figure_pT_us_PPPdiff.addtext(t) DpT_us_PPP (Singapore):  1970-2006
figure_pT_us_PPPdiff.addtext(b) Year
figure_pT_us_PPPdiff.addtext(l) DpT_us_PPP
figure_pT_us_PPPdiff.legend(off)

'The graph is not particularly illuminating.  Depending on how you look at it, it could have a time trend to it.  However, since the previous ADF test concluded that the equation for pT_us_PPP do have a time trend, we run the ADF test with both a constant as well as a time trend.

'So we begin the whole process over again: 

genr pT_us_PPPdiff = d(pT_us_PPP)
freeze(table_8_2_pT_us_PPPdiff1_adf) pT_us_PPPdiff.uroot(adf,const,info=sic)

'Note that the SIC automatic lag selection picks no lags, p =0.  The unit root test produces a t-value of -3.87 which is smaller our 5% criterion -2.95.  Thus, we may now reject the null of nonstationarity in first differenced series of pT_us_PPP.  There is no reason to go further.  The last thing we do is check for white noise.

genr resid = 0
freeze(mode=overwrite,pT_us_PPPdiff1_adf) pT_us_PPPdiff.uroot(adf,const,trend,info=sic)
freeze(pT_us_PPPdiff1_adf_correl) resid.correl

''Based on the Q-statistic, I conclude that the residuals are white noise.  Putting it all together, I conclude that the pT_us_PPP series is I(1).

'*********************************************************************************
'DF-GLS Unit Root Test for Singapore's pT_us_PPP
'We now run our first dfgls test
'**********************************************************************************
 
freeze(table_8_2a_pT_us_PPP_dfgls) pT_us_PPP.uroot(dfgls,trend,info=sic)

'Note that the SIC automatic lag selection picks lags, p = 1.  The unit root test produces a t-value of -1.50 which is greater than our 5% criterion -3.19.  Thus, at this point, we may not reject the null of a unit root.

'Now let's see if the series is difference stationary or not

genr pT_us_PPPdiff = d(pT_us_PPP)
freeze(table_8_2b_pT_us_PPPdiff1_dfgls) pT_us_PPPdiff.uroot(dfgls,const,info=sic)

'Note that the SIC automatic lag selection picks no lags, p = 0.  The unit root test produces a t-value of -3.82 which is smaller than our 5% criterion -1.95.  Thus, we may now reject the null of nonstationarity in first differenced series of pT_us_PPP.  
 
''Putting it all together, I conclude that the pT_us_PPP series is I(1), a finding compatable with my ADF test results.

''******************************************************************************************
''S1: Establishing Cointegration for Verifying PPP for Inter-Country Tradables Prices
'******************************************************************************************
'Now, by employing FMOLS and DOLS cointegration regression estimators, we shall calculate our LR cointegrating vectors.

equation table_8_2_LRPPP_fmols.cointreg(method=fmols) pT pT_us_PPP

equation table_8_2_LRPPP_dols.cointreg(method=dols, trend=constant, lag=1,lead=1 ) pT pT_us_PPP

'As proven by FMOLS and DOLS test results, valid cointegration holds between pT and pT_us_PPP. Thus, I can proceed with estimation of S2 condition. 

''************************************************************************************************
''S2: Testing for Equi-proportionate Relationship between Inter-Country Tradables Prices
'************************************************************************************************

freeze(table_8_2_LRPPP_fmols_Ftest) table_8_2_LRPPP_fmols.wald c(1)=1

freeze(table_8_2_LRPPP_dols_Ftest) table_8_2_LRPPP_dols.wald c(1)=1

'As proven by Wald coefficient test results for FMOLS estimates (only), PPP does not hold validly for traded sector prices of Singapore and U.S. Thus, there is a 'MIXED' support for the existence of PPP for Singapore and U.S. tradables prices.So, I will proceed with estimating the modified version of BS hypothesis. 

'***************************************************************************************
'ESTIMATING BALASSA-SAMUELSON EFFECT FOR rer_def_NT, rer_def_T  & a_tilde
'**************************************************************************************
'*************************************************************
'STEP 0: Tests for Unit Root in Individual Time Series
'*************************************************************
'***************************************************************
'Graph for Singapore's  rer_def_NT
'***************************************************************
                                        
genr rer_def_NT = rer_def_NT
freeze(figure_rer_def_NT) rer_def_NT.line
figure_rer_def_NT.addtext(t) rer_def_NT (Singapore):  1970-2006
figure_rer_def_NT.addtext(b) Year
figure_rer_def_NT.addtext(l) rer_def_NT
figure_rer_def_NT.legend(off)
                                                 
'We see from the FIGURE that rer_def_NT has a time trend to it.  So we would include both an intercept and a time trend in our unit root regression equtions. 

'*******************************************************
'ADF Unit Root Test for Singapore's RER_DEF_NT
'*******************************************************
 
freeze(table_8_7_rer_def_nt_adf) rer_def_nt.uroot(adf,trend,info=sic)

'Note that the SIC automatic lag selection picks lags, p = 1. The unit root test produces a t-value of -3.56 which is smaller than our 5% criterion -3.54.  Thus, at this point, we can reject the null of a unit root.

'Now, let's check for white noise. To do that, I first set all the residuals = 0, then run the ADF test and finally will check for white noise.

genr resid = 0
freeze(mode=overwrite,rer_def_nt_adf) rer_def_nt.uroot(adf,const,trend,info=sic)
freeze(rer_def_nt_adf_correl) resid.correl
 
'Based on the Q-statistic, I conclude that the residuals are white noise.  Putting it all together, I conclude that the rer_def_nt series is not level stationary.

'**************************************************************
'DF-GLS Unit Root Test for Singapore's RER_DEF_NT
'**************************************************************
 
freeze(table_8_7_rer_def_nt_dfgls) rer_def_nt.uroot(dfgls,trend,info=sic)

'Note that the SIC automatic lag selection picks lags, p = 1. The unit root test produces a t-value of -3.52 which is smaller than our 5% criterion -3.19. Thus, at this point, we may reject the null of a unit root. 

''Putting it all together, I conclude that the rer_def_nt series is I(0), a finding compatible with my ADF test results.

'*************************************************
'Graph for Singapore's Productivity (a_tilde)
'*************************************************
                                        
genr a_tilde = a_tilde
freeze(figurea_tilde) a_tilde.line
figurea_tilde.addtext(t) a_tilde (Singapore):  1970-2006
figurea_tilde.addtext(b) Year
figurea_tilde.addtext(l) a_tilde
figurea_tilde.legend(off)
                                                 
'We see from the FIGURE that a_tilde has time trend to it.  So we would include both an intercept and a time trend in our unit root regression equations. 

'*******************************************************
'ADF Unit Root Test for Singapore's Productivity
'*******************************************************
 
freeze(table_8_7_a_tilde_adf) a_tilde.uroot(adf,trend,info=sic)

'Note that the SIC automatic lag selection picks lags, p =1.  The unit root test produces a t-value of -3.34 which is greater than our 5% criterion -3.52.  Thus, at this point, we can reject the null of a unit root.

'Now, let's check for white noise.  To do that, I first set all the residuals = 0, then run the ADF test and finally will check for white noise.

genr resid = 0
freeze(mode=overwrite,a_tilde_adf) a_tilde.uroot(adf,const,trend,info=sic)
freeze(a_tilde_adf_correl) resid.correl
 
'Based on the Q-statistic, I conclude that the residuals are white noise.  Putting it all together, I conclude that the a_tilde series is not level stationary.

'The next thing I do is test whether the differenced series is stationary using the ADF test.  I once again begin by graphing the (differenced) series.
 
genr a_tildediff = d(a_tilde)
freeze(figure_a_tildediff) a_tildediff.line
figure_a_tildediff.addtext(t) da_tilde (Singapore):  1970-2006
figure_a_tildediff.addtext(b) Year
figure_a_tildediff.addtext(l) Da_tilde
figure_a_tildediff.legend(off)

'From the graph, the series clearly does not have a time trend to it. So, I would test the series for unit with an intercept only.

'So we begin the whole process over again: 

genr a_tildediff = d(a_tilde)
freeze(table_8_7_a_tildediff1_adf) a_tildediff.uroot(adf,const,info=sic)

'Note that the SIC automatic lag selection picks no lags, p =0.  The unit root test produces a t-value of -3.46 which is now smaller than our 5% criterion -2.95.  Thus, we may now reject the null of non-stationarity in first differenced series of a_tilde.  There is no reason to go further.  The last thing we do is to check ADF regression result for white noise.

genr resid = 0
freeze(mode=overwrite,a_tildediff1_adf) a_tildediff.uroot(adf,const,info=sic)
freeze(a_tildediff1_adf_correl) resid.correl

''Based on the Q-statistic, I conclude that the residuals are white noise.  Putting it all together, I conclude that the a_tilde series is I(1).

'************************************************************
'DF-GLS Unit Root Test for Singapore's Productivity
'************************************************************
 
freeze(table_8_7_a_tilde_dfgls) a_tilde.uroot(dfgls,trend,info=sic)

'Note that the SIC automatic lag selection picks lags, p = 1.  The unit root test produces a t-value of -3.45 which is smaller than our 5% criterion -3.19.  Thus, at this point, we may  reject the null of a unit root.  

''Putting it all together, I conclude that the a_tilde series is I(0), a finding incompatible with my ADF test results.

'***************************************************************
'Graph for Singapore's  rer_def_T
'***************************************************************
                                        
genr rer_def_T = rer_def_T
freeze(figure_rer_def_T) rer_def_T.line
figure_rer_def_T.addtext(t) rer_def_T (Singapore):  1970-2006
figure_rer_def_T.addtext(b) Year
figure_rer_def_T.addtext(l) rer_def_T
figure_rer_def_T.legend(off)
                                                 
'We see from the FIGURE that rer_def_T has a time trend to it.  So we would include both an intercept and a time trend in our unit root regression equtions. 

'*********************************************************************************
'ADF Unit Root Test for Singapore's rer_def_T
'We now run our first ADF test
'**********************************************************************************
 
freeze(table_8_7_rer_def_T_adf) rer_def_T.uroot(adf,trend,info=sic)

'Note that the SIC automatic lag selection picks lags, p = 1.  The unit root test produces a t-value of -3.77 which is smaller than our 5% criterion -3.54.  Thus, at this point, we can reject the null of a unit root.

'Now, let's check for white noise.  To do that, I first set all the residuals = 0, then run the adf test and finally will check for white noise.

genr resid = 0
freeze(mode=overwrite,rer_def_T_adf) rer_def_T.uroot(adf,const,trend,info=sic)
freeze(rer_def_T_adf_correl) resid.correl
 
'Based on the Q-statistic, I conclude that the residuals are white noise. Putting it all together, I conclude that the rer_def_T series is level stationary.

'*********************************************************************************
'DF-GLS Unit Root Test for Singapore's rer_def_T
'We now run our first dfgls test
'**********************************************************************************
 
freeze(table_8_7_rer_def_T_dfgls) rer_def_T.uroot(dfgls,trend,info=sic)

'Note that the SIC automatic lag selection picks lags, p = 1.  The unit root test produces a t-value of -3.31 which is smaller than our 5% criterion -3.19.  Thus, at this point, we may reject the null of a unit root. 
 
''Putting it all together, I conclude that the rer_def_T series is I(0), a finding compatable with my ADF test results.

'*********************************************
'Single Equation Cointegration Methods
'*********************************************

'******************************************************************
''Graph the suspected cointegrated series together
'******************************************************************
'The first step is to print out a graph of the series.  This is very important!

group g1 rer_def_NT rer_def_T a_tilde
freeze(figure1) g1.line(x)
figure1.setelem(1) lcolor(black) symbol(1) lpat(1)
figure1.setelem(2) lcolor(black) symbol(4) lpat(1)
figure1.setelem(3) lcolor(black) symbol(7) lpat(1)
figure1.setelem(3) lcolor(black)
figure1.options linepat
figure1.addtext(t) rer_def_NT, lnpT & a_tilde (Singapore & U.S): 1970-2006
figure1.addtext(b) Year
figure1.addtext(l) rer_def_NT
figure1.addtext(l) lnpT
figure1.addtext(r) a_tilde

''*******************************************************
''S1.A.Engle-Granger Approach to Cointegration
'*******************************************************

genr resid = 0
equation eg.ls rer_def_NT c rer_def_T a_tilde
genr EC1 = resid

'First we test if the residuals of above regression are level stationary or not. If yes, next we'll proceed towards estimation of error correction model.

'***************************************************************
'Graph for Singapore's  EC
'***************************************************************                                               
genr EC1 = EC1
freeze(figure_EC1) EC1.line
figure_EC1.addtext(t) EC1 (Singapore):  1970-2006
figure_EC1.addtext(b) Year
figure_EC1.addtext(l) EC1
figure_EC1.legend(off)
                                                 
'We see from the FIGURE that EC has time trend to it.  So we would include both an intercept and trend in our unit root regression equtions. 

'*********************************************************************************
'EG Test for Cointegration
'**********************************************************************************
 
freeze(table_8_7_EGC) g1.coint(method=eg)

'The null hypothesis will not be rejected as suggested by sample statistics.

''******************************************
''S1.B.Error Correction Model (ECM)
'*******************************************
''*******************************************************
'Selecting the number of lags in the VAR  *
'*******************************************************
'NOTE: We do this because we need to have the "right" number of lags when it comes time to estimate our VEC model and test for cointegration.

var var1.ls 1 6   g1
freeze(var1_lagtest1) var1.laglen(6)
freeze(var1_lagtest2) var1.testlags

'The laglength test above indicates that the should have VAR has 2.

var var2.ls 1 2  g1
freeze(var2_artest1) var2.correl
freeze(var2_artest2) var2.qstats(12)
freeze(var2_artest3) var2.arlm(12)

'We now try different lags of d(rer_def_T) and d(a_tilde), comparing SIC values across specifications.

genr resid = 0
equation eg.ls rer_def_NT c rer_def_T a_tilde
genr ec1 = resid

var table_8_7_eg2a.ls 0 0 d(rer_def_NT)   @  c ec1(-1) d(rer_def_NT(-1)) d(rer_def_NT(-2)) 

var table_8_7_eg2b.ls 0 0 d(rer_def_NT)   @  c ec1(-1) d(rer_def_NT(-1)) d(rer_def_NT(-2)) d(rer_def_T(-1)) d(a_tilde(-1))

var table_8_7_eg2c.ls 0 0 d(rer_def_NT)   @  c ec1(-1) d(rer_def_NT(-1)) d(rer_def_NT(-2)) d(rer_def_T(-1)) d(rer_def_T(-2)) d(a_tilde(-1)) d(a_tilde(-2))

'The evidence suggests that Model B is best.  Now we test that model for serial correlation.

var table_8_7_eg2b.ls 0 0 d(rer_def_NT)   @   c ec1(-1) d(rer_def_NT(-1)) d(rer_def_NT(-2)) d(rer_def_T(-1)) d(a_tilde(-1))
freeze(table_8_7_eg2b_artest1) table_8_7_eg2b.correl
freeze(table_8_7_eg2b_artest2) table_8_7_eg2b.qstats(12)
freeze(table_8_7_eg2b_artest3) table_8_7_eg2b.arlm(12)

'The residuals are absolutely white noise.

''*************************
''Estimating EC Model  
'**************************

'We'll now take the above specified model and turn it into an ECM. We shall run NW-HAC least squares model for establishing error correction mechanism.

'We now estimate the corresponding ECM:

equation table_8_7_ecm.ls(n) d(rer_def_NT) c ec1(-1) d(rer_def_NT(-1)) d(rer_def_NT(-2)) d(rer_def_T(-1)) d(a_tilde(-1))

'Note that the SR effect is insignificant as the EC coefficient is of value (0.05) is statistically insignificant even at 10% significance level besides holding an undesired sign (+).

''******************************************
'Multivariate Cointegration Approach
'******************************************

''*************************************************************
''Check if the VAR(2) model is dynamically stable
'*************************************************************
freeze(table_8_7_var2_varstable) var2.arroots(graph)

'The model is dynamically stable.

''**********************************************************************
''M1.A & M1.B: Identifying the number of cointegrating vectors
'***********************************************************************
'Having identified the appropriate number of lags to put in, I now go on to test for the appropriate number of cointegrating equations.

freeze(table_8_7_var2_coint) var2.coint(s,2)

'This command estimates all possible combinations of constants and trends in the level data series and the cointegrating equations. All the results indicate 1 cointegrating vectors.
'
'GENERAL NOTE:, in practice, cases 1 and 5 are rarely used. One should use case 1 only if one knows that all series have zero mean. Case 5 may provide a good fit in-sample but will produce implausible forecasts out-of-sample. As a rough guide, use case 2 if none of the series appear to have a trend. For trending series, use case 3 if you believe all trends are stochastic; if you believe some of the series are trend stationary, use case 4.

'Note that the 5 cases are identified under "Johansen cointegration test" in Eviews. They run from most restrictive (no constants in either the level series or CEs) to most general (trend terms in both the level series and CEs).

'CONCLUSION:  I conclude that rer_def_NT and a_tilde are not cointegrated in the Singapore's data.


